/*
 * Copyright 2006-2007 the original author or authors.
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *      http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
package asphalt2.util
{
	import flash.display.DisplayObject;
	import flash.events.Event;
	import flash.events.IEventDispatcher;
	import flash.geom.Point;
	
	import mx.core.Application;
	import mx.core.Container;
	
	/**
	 * UI用ユーティリティ
	 * @author c9katayama
	 */
	public class UIUtil
	{
		/**
		 * targetが自分か自分の親オブジェクトかどうかをチェックします。
		 **/	
		public static function isParentOrThis(comp:DisplayObject,target:DisplayObject):Boolean{
			if(comp==target)
				return true;
			return isParent(comp,target);
		}
		/**
		 * targetが自分の親オブジェクトかどうかをチェックします。
		 **/	
		public static function isParent(comp:DisplayObject,target:DisplayObject):Boolean{
			if(!(target is Container)){
				return false;
			}
			var _parent:DisplayObject = comp.parent;
			while(true){
				if(_parent == null){
					break;
				}else{
					if(_parent == target){
						return true;
					}
				}
				_parent = _parent.parent;
			}
			return false;
		}
		/**
		 * 引数のコンポーネントの一番上の親を返します。
		 * 引数がnullの場合、Application.applicationから検索を行います。 
		 **/
		public static function findRoot(component:DisplayObject=null):DisplayObject
		{
			if(component==null){
				component = Application.application as DisplayObject;
			}
			var parent:DisplayObject = component;
			while(true){
				if(parent.parent==null){
					break;					
				}else{
					parent = parent.parent;
				}
			}
			return parent;
		}
		/**
		 * 1回のみ実行するイベントリスナーを追加します。 
		 **/
		public static function addOnceEventListener(dispatcher:IEventDispatcher,eventType:String,listener:Function):void{
			EventUtil.addOnceEventListener(dispatcher,eventType,listener);
		}
		/**
		 * 1回のみ実行するイベントリスナーをremoveします。 
		 **/
		public static function removeOnceEventListener(dispatcher:IEventDispatcher,eventType:String,listener:Function):void{
			EventUtil.removeOnceEventListener(dispatcher,eventType,listener);
		}
		/**
		 * 対象とするDisplayObjectがRemoveされた時に、listenerをremoveします。
		 */
		public static function addComponetScopeEventListener(dispatcher:IEventDispatcher,lifeCycleComponent:DisplayObject,type:String,listener:Function):void{
			dispatcher.addEventListener(type,listener);
			UIUtil.callBackWhenComponentRemoved(lifeCycleComponent,function():void{
				ApplicationEventDispatcher.removeEventListener(type,listener);
			});
		}		
		/**
		 * 対象コンポーネントもしくは親コンポーネントがRemoveされた時に、callBackFunctionを呼び出します。
		 */
		public static function callBackWhenComponentRemoved(component:DisplayObject,callBackFunction:Function):void{
			var root:DisplayObject = findRoot(component);
			var parentRemoveFunction:Function = function(e:Event):void{
				if(UIUtil.isParentOrThis(component,e.target as DisplayObject)){
					root.removeEventListener(Event.REMOVED,parentRemoveFunction);
					callBackFunction();
					parentRemoveFunction=null;						
				}
			};
			root.addEventListener(Event.REMOVED,parentRemoveFunction);
		}
		/**
		 * x,yトを表示オブジェクトの (ローカル) 座標からステージ (グローバル) 座標に変換します。 
		 **/
		public static function localToGlobal(component:DisplayObject,x:Number=0,y:Number=0):Point{
			var p:Point = new Point(x,y);
			p = component.localToGlobal(p);
			return p;
		}
		/**
		 * x,yをステージ (グローバル) 座標から表示オブジェクトの (ローカル) 座標に変換します。
		 */
		public static function globalToLocal(component:DisplayObject,x:Number=0,y:Number=0):Point{
			var p:Point = new Point(x,y);
			p = component.globalToLocal(p);
			return p;
		}
	}
}